
	Reason for change:
	The definition of the 3gpp-Sbi-Consumer-Info header contains the following ABNF rule:

servicename = 1*( "-" / %x61-7A )

which allows characters "a-z" (only lower case) and the "-" character.

This is not enough to describe service names that contain digits, such as:

- nnrf-oauth2
- n5g-eir-eic
- n5gddnmf-discovery
- nmfaf-3dadm
- nmfaf-3cadm

In addition, in the definition of the scope attribute in Oauth2 access tokens (which are composed by using the service name as prefix), the "A-Z" (upper case) characters, and the "_" character, have been allowed since Rel-15; although these characters have never been used by 3GPP standard services, they might be used in custom services.

scope:
  type: string
  pattern: '^([a-zA-Z0-9_:-]+)( [a-zA-Z0-9_:-]+)*$'

So, it seems safer to re-define the "servicename" rule by allowing (in addition to the existing characters):
- "A-Z" (upper case)
- "0-9" (digits)
- "_" character

* * * First Change * * * *
[bookmark: _Toc145654412]5.2.3.3.7	3gpp-Sbi-Consumer-Info
This header contains a comma-delimited list of NF service consumer information from an HTTP client (as NF service consumer).
The encoding of the header follows the ABNF as defined in IETF RFC 9110 [11].

Sbi-Consumer-Info-Header = "3gpp-Sbi-Consumer-Info:" OWS 1#( supportedService ";" OWS supportedVersions [ ";" OWS supportedFeatures ] [ ";" OWS acceptEncoding ] [ ";" OWS callback-uri-prefix ] [ ";" OWS intraPlmnCallbackRoot ";" OWS interPlmnCallbackRoot ] ) OWS

supportedService = "service=" servicename

servicename = 1*( "-" / %x30-39 / %x41-5A / "_" / %x61-7A )

supportedVersions = "apiversion=" "(" OWS [ apimajorversion *( RWS apimajorversion ) OWS ] ")"

apimajorversion = %x31-39 [ *DIGIT ]

supportedFeatures = "supportedfeatures=" features

features = *HEXDIG

acceptEncoding = "acceptencoding=" %x22 encodingList %x22

encodingList = #( codings [ weight ] )

intraPlmnCallbackRoot = "intraPlmnCallbackRoot=" DQUOTE sbi-scheme "://" sbi-authority
                        [ prefix ] DQUOTE

interPlmnCallbackRoot = "interPlmnCallbackRoot=" DQUOTE sbi-scheme "://" sbi-authority
                        [ prefix ] DQUOTE

sbi-scheme = "http" / "https"

sbi-authority = host [ ":" port ]

port = *DIGIT

prefix = path-absolute ; path-absolute production rule from IETF RFC 3986, clause 3.3

callback-uri-prefix = "callback-uri-prefix=" DQUOTE prefix DQUOTE


"service" (Mandatory parameter): Supported Service parameter indicates the name of a service, as defined in 3GPP TS 29.510 [8], which is supported by the sender as NF service consumer.
"apiversion" (Mandatory parameter): Supported Versions parameter indicates the major version(s) of the service API that are supported by the sender as NF service consumer.
"supportedfeatures" (Optional parameter): Supported Features parameter carries a string containing a bitmask in hexadecimal representation, as specified for SupportedFeatures data type in 3GPP TS 29.571 [13], to indicate the feature(s) of the service API that are supported by the sender as NF service consumer.
"acceptencoding" (Optional parameter): Accept Encoding carries a string indicating the accepted content encodings supported by the sender as NF service consumer, when receiving notifications defined by the service. In the ABNF definition, "codings" and "weight" are defined in IETF RFC 9110 [11] clauses 12.5.3 and 12.4.2.
"intraPlmnCallbackRoot", "interPlmnCallbackRoot" (Optional parameters): intra plmn callback root and inter plmn callback root supported by the sender as NF service consumer, for the indicated service.
"callback-uri-prefix" (Optional parameter): The NF service consumer may include this parameter when providing a Callback URI when the authority part of the Callback URI is shared by several NF service consumer instances. The NF service consumer may also include this header when providing a Callback URI including a prefix, for use during NF service consumer reselection, when binding procedures are not supported. When present, the "callback-uri-prefix" shall be a path-absolute as specified IETF RFC 3986 [14] (i.e. the first path segment(s) after the authority) which is part of the Callback URI provided by a NF service consumer in the corresponding service request message sent to a NF service producer. The authority and "callback-uri-prefix" in the Callback URI shall uniquely identify a consumer service instance. See clause 6.12.1 for the usage of this parameter.
EXAMPLE 1:	The NF consumer supports Namf_EventExposure OpenAPI "v1" without any optional feature:
	3gpp-Sbi-Consumer-Info: service=namf-evts; apiversion=(1)
EXAMPLE 2:	The NF consumer supports Nsmf_EventExposure OpenAPI "v1" and "v2" with optional feature number 1 and accepted encoding "gzip":
	3gpp-Sbi-Consumer-Info: service=nsmf-event-exposure; apiversion=(1 2); supportedfeatures=01; acceptencoding="gzip; q=1.0, *;q=0.5"
EXAMPLE 3:	The NF consumer supports both Namf_EventExposure OpenAPI "v1" and Nsmf_EventExposure OpenAPI "v2":
	3gpp-Sbi-Consumer-Info: service=namf-evts; apiversion=(1), service=nsmf-event-exposure; apiversion=(2)
[bookmark: _PERM_MCCTEMPBM_CRPT57490061___5]EXAMPLE 4:	An AMF service instance supports Nsmf_PDUSession OpenAPI "v1", provides the callback URI https://amf45.operator.com/servinst123/pdusession, whereby the authority is shared by more than one AMF service instance, while the prefix "/servinst123" uniquely identifies a specific AMF service instance:

3gpp-Sbi-Consumer-Info: service=nsmf-pdusession; apiversion=(1); callback-uri-prefix="/servinst123"
[bookmark: _PERM_MCCTEMPBM_CRPT57490062___5]EXAMPLE 5:	The NF consumer supports Namf_EventExposure OpenAPI "v1" and sends intra PLMN callback root "https://operator.com" and inter PLMN callback root "https://5gc.mnc012.mcc345.3gppnetwork.org" in the header:
	3gpp-Sbi-Consumer-Info: service=namf-evts; apiversion=(1); intraPlmnCallbackRoot="https://operator.com"; interPlmnCallbackRoot="https://5gc.mnc012.mcc345.3gppnetwork.org"

* * * Next Change * * * *
[bookmark: _Toc145654605]D.2	ABNF definitions (Filename: "TS29500_CustomHeaders.abnf")

; ----------------------------------------
;   RFC 5234
; ----------------------------------------


(… text not shown for clarity …)


; ----------------------------------------
;   3GPP TS 29.500
;
;   Version: 18.34.0 (SeptDecember 2023)
;
;   (c) 2023, 3GPP Organizational Partners (ARIB, ATIS, CCSA, ETSI, TSDSI, TTA, TTC).
; ----------------------------------------


;
; Header: 3gpp-Sbi-Message-Priority
;


(… text not shown for clarity …)


;
; Header: 3gpp-Sbi-Consumer-Info
;

Sbi-Consumer-Info-Header = "3gpp-Sbi-Consumer-Info:" OWS
                           consumer-info-element *( OWS "," OWS consumer-info-element ) OWS

consumer-info-element    = ( supportedService ";" OWS supportedVersions 
                             [ ";" OWS supportedFeatures ]
                             [ ";" OWS acceptEncoding ]
                             [ ";" OWS callback-uri-prefix ]
                           ) [ ";" OWS intraPlmnCallbackRoot ";" OWS interPlmnCallbackRoot ]

supportedService         = "service=" servicename

servicename              = 1*( "-" / %x30-39 / %x41-5A / "_" / %x61-7A )

supportedVersions        = "apiversion=" "(" OWS
                           [ apimajorversion *( RWS apimajorversion ) OWS ] ")"

apimajorversion          = %x31-39 [ *DIGIT ]

supportedFeatures        = "supportedfeatures=" features

features                 = *HEXDIG

acceptEncoding           = "acceptencoding=" %x22 encodingList %x22

encodingList             = [ encoding-element *( OWS "," OWS encoding-element ) ]

intraPlmnCallbackRoot    = "intraPlmnCallbackRoot="
                           DQUOTE sbi-scheme "://" sbi-authority [ prefix ] DQUOTE

interPlmnCallbackRoot    = "interPlmnCallbackRoot="
                           DQUOTE sbi-scheme "://" sbi-authority [ prefix ] DQUOTE

callback-uri-prefix      = "callback-uri-prefix=" DQUOTE prefix DQUOTE


(… text not shown for clarity …)

* * * End of Changes * * * *
